/*
 *  Copyright (c) 2010 Brian Stanek <bstanek@gmail.com>
 *
 *  Permission is hereby granted, free of charge, to any person
 *  obtaining a copy of this software and associated documentation
 *  files (the "Software"), to deal in the Software without
 *  restriction, including without limitation the rights to use,
 *  copy, modify, merge, publish, distribute, sublicense, and/or sell
 *  copies of the Software, and to permit persons to whom the
 *  Software is furnished to do so, subject to the following
 *  conditions:
 *
 *  The above copyright notice and this permission notice shall be
 *  included in all copies or substantial portions of the Software.
 *
 *  THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND,
 *  EXPRESS OR IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES
 *  OF MERCHANTABILITY, FITNESS FOR A PARTICULAR PURPOSE AND
 *  NONINFRINGEMENT. IN NO EVENT SHALL THE AUTHORS OR COPYRIGHT
 *  HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER LIABILITY,
 *  WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING
 *  FROM, OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR
 *  OTHER DEALINGS IN THE SOFTWARE.
 */

package guice.binding
{

import guice.Binder;
import guice.Binding;
import guice.Injector;
import guice.Key;
import guice.Provider;
import guice.provider.InstanceProvider;
import guice.scopes.Scoping;
import guice.spi.BindingTargetVisitor;
import guice.spi.InstanceBinding;

public class InstanceBindingImpl extends AbstractBinding
        implements InstanceBinding
{

    //--------------------------------------------------------------------------
    //
    //  Constructor
    //
    //--------------------------------------------------------------------------

    public function InstanceBindingImpl(source:Object, key:Key, scoping:Scoping,
            instance:*)
    {
        super(source, key, scoping);
        this.instance = instance;
        provider = new InstanceProvider(instance);
    }

    //--------------------------------------------------------------------------
    //
    //  Variables
    //
    //--------------------------------------------------------------------------

    private var instance:*;

    private var provider:Provider;

    //--------------------------------------------------------------------------
    //
    //  Methods: InstanceBinding
    //
    //--------------------------------------------------------------------------

    public function getInstance():*
    {
        return instance;
    }

    //--------------------------------------------------------------------------
    //
    //  Overridden Methods
    //
    //--------------------------------------------------------------------------

    override public function acceptTargetVisitor(visitor:BindingTargetVisitor):Boolean
    {
        return visitor.visitInstanceBinding(this);
    }

    override public function applyTo(binder:Binder):void
    {
        binder.withSource(getSource()).bindKey(getKey()).toInstance(instance);
    }

    override public function getProvider():Provider
    {
        return provider;
    }

    override public function withKey(key:Key):AbstractBinding
    {
        return new InstanceBindingImpl(getSource(), key, getScoping(),
                instance);
    }

    override public function withScoping(scoping:Scoping):AbstractBinding
    {
        return new InstanceBindingImpl(getSource(), getKey(), scoping,
                instance);
    }

}

}
